UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

A Program for Mass Spectrometer Control and Data Processing Analyses in Isotope Geology--Written in BASIC for an 8K Nova 1210 Computer

By

J. S. Stacey
U.S. Geological Survey

and

J. Hope
Bernard Price Institute of Geophysics
University of Witwatersrand
Johannesburg, South Africa

Open-file report 75-127
1975

This report is preliminary and has not been edited or reviewed for conformity with U.S. Geological Survey standards and nomenclature.

CONTENTS

Illustrations

Tables

A PROGRAM FOR MASS SPECTROMETER CONTROL AND DATA PROCESSING ANALYSES IN ISOTOPE GEOLOGY--WRITTEN IN BASIC FOR AN 8K NOVA 1210 COMPUTER

By

J. S. Stacey
U.S. Geological Survey

and

J. Hope
Bernard Price Institute of Geophysics
University of Witwatersrand
Johannesburg, South Africa

ABSTRACT

A system is described which uses a minicomputer to control a surface ionization mass spectrometer in the peak switching mode, with the object of computing isotopic abundance ratios of elements of geologic interest. The program uses the BASIC language and is sufficiently flexible to be used for multiblock analyses of any spectrum containing from two to five peaks. In the case of strontium analyses, ratios are corrected for rubidium content and normalized for mass spectrometer fractionation. Although almost any minicomputer would be suitable, the model used was the Data General Nova 1210 with 8K memory. Assembly language driver program and interface hardware descriptions for the Nova 1210 are included.

INTRODUCTION

This program was written while J. S. Stacey was visiting the Bernard Price Institute of Geophysics at the University of Witwatersrand in Johannesburg, South Africa. Although there was not sufficient operational time available to incorporate all possible levels of sophistication, the program does offer a very flexible and comprehensive package which could be used by almost any minicomputer. This article may therefore be of interest to people in other isotope geology laboratories. The main concepts were developed over a period of several years at the U.S. Geological Survey's laboratory in Denver, Colorado, using a variety of instruments and computer systems and we particularly wish to acknowledge the work of R. Terrezas and E. E. Wilson in this regard. The program described here is, however, the first we have written in BASIC language--which is well suited for laboratory use. Changes in BASIC are very easy to make and the format is easy to understand. The version of BASIC used "BASIC with CALL package," was written for the Data General Nova 1210, by Perseus Computing and Automation (Pty) ltd., P. O. Box 3987, Pretoria, South Africa. It facilitates communication between BASIC and Nova's assembly language and considerably simplified the writing of the driver program to control the mass spectrometer. The program will handle any spectrum containing up to five peaks in a switching mode. Included therefore are analyses of uranium, thorium, rubidium, lead, and strontium. In the case of strontium, corrections are made for rubidium and the ratios are normalized for fractionation.

Features which could easily be incorporated are "peak top and baseline filtering" and the facility to reject data during an analysis (Stacey and others, 1971; 1972). Sensitivity range switching for the vibrating reed electrometer could easily be added if desired.

Actually, since this report was initiated, the program described here has been further developed and adapted to run on the Hewlett Packard HP9830 calculators now used in Denver for mass spectrometer control. Some simplification has resulted by writing the strontium option as a separate program from that for the other elements. The new listings are available from our Denver laboratory.

These new programs have retained the structure of the original program, and rather than outmoding the material in this report they serve to emphasize its fundamental simplicity and flexibility.

MASS SPECTROMETER CONTROL SYSTEM

The mass spectrometer at the Bernard Price Institute of Geophysics is provided with magnetic field control by a Bell Inc., type 640, incremental gaussmeter. 5 independent peak channels are offered by 5 potentiometers. Each potentiometer can be selected separately by its own relay operated from the computer. Two other potentiometers are provided so that magnetic field offsets can be adjusted for all peaks simultaneously. These offset positions are selected by two relays and correspond to "baseline below peak" (offset reduces field), "on peak" (zero offset), "baseline above peak" (offset increases field) respectively. The circuitry arrangement for this is shown in Appendix D. Altogether, a total of 7 relays are driven from the computer--5 for the peaks and 2 for the offsets.

Analog signals from the Carey Model 31 vibrating reed electrometer (1 volt for full scale), are measured by a voltage-to-frequency converter and counter combination (Hewlett Packard HP221 B and HP5326A respectively). Normally signals are integrated over a period of one second. Therefore almost once per second the computer is required to read a six-character BCD measurement from the counter. The A-register on the Nova general purpose interface board is a 16-bit device and is arranged to read the 4 least significant digits. The 2 most significant digits are read by 8 of the 12 bits of the B-register.

The C-register forms the "interblock print-time counter" which counts the number of measurements made by the counter(that is, the number of seconds taken) during data printout between blocks in a mass spectrometer analysis. Both the B and C registers were assembled on the Nova general purpose interface board.

FUNCTIONS OF THE PROGRAM

Any type of analysis involving up to five peaks may be made. Therefore analyses of U, Th, Rb, Pb, and Sr are included. In analyses where three or more peaks are measured, #3 peak is always the reference peak--that is, all ratios are measured with respect to this peak. Thus, in a lead analysis for instance, Pb^{206} may conveniently be assigned to #3 channel and the others as below:

^{204}Pb to #1 or #2 channels

^{206}Pb to #3 channel

^{207}Pb to #4 channel

^{208}Pb to #5 channel

For strontium analyses it is important to assign the isotopes as indicated below because of the ^{85}Rb correction and the normalizing computations.

^{84}Sr to #1 or #2 channels

^{85}Rb to #2 or #1 channels

^{86}Sr to #3 channel

^{87}Sr to #4 channel

^{88}Sr to #5 channel

In strontium analyses the ^{85}Rb correction peak is measured only at the beginning and end of each block--as are the baselines for each of the peaks in the spectrum.

A block comprises baseline measurements for each peak taken below only, above only, or above and below each peak as requested by the operator. For strontium analyses the ^{85}Rb peak is then measured with its 'below' baseline. This is followed by measurement of the main peaks switched through in sequence as many times as has been requested by the operator at the beginning of the analysis. In strontium analyses, after the main peak switching, the ^{85}Rb peak and its baseline are measured and the block is ended with a final set of baselines for each of the main peaks in the same sequence as at the beginning of the block.

If more than one block is to be measured, then while the teletype is printing ratios at the end of a block, the C-register built into the interface keeps track of time. Time is clocked by the number of measurements made by the main signal measuring counter (at the rate of one per second in our case). Time for the printout between blocks is read from the C-register and is added to the total time elapsed at the start of the next block of data. This enables the baselines at the end of the nth block to be used as the first baselines in the (n + 1)th block and so on throughout the analysis.

The time spent on each peak (that is, the number of measurements made at one per second) can be different for each peak, as also can be the times (skip times) allowed for switching transients at the beginning of each peak. Each baseline measurement is given 8 seconds longer than its own peak measurement and 2 extra seconds are skipped to allow switching transients to fully decay. Similarly, extra periods are given for the ^{85}Rb peak over those ascribed to the ^{87}Sr peak.

As far as the computer is concerned there is no limit to the time which can be requested for a peak--but blocks may not consist of more than 10 switching cycles through the main peaks and an analysis may not comprise more than 10 blocks.

Between blocks, individual ratios and their times are printed, as also are their means and standard deviations. The data are filtered by rejecting ratios more than 2 standard deviations from the mean. The rejected ratios and times are printed, followed by the new mean and standard deviation. The process is repeated until no ratios remain outside the 2 sigma limit.

At the end of an analysis a table is printed of all the mean ratios for the blocks and the table ends with the final means and standard deviations for the whole analysis.

For strontium analyses, rubidium corrections are made for each block. Because of memory storage limitations with an 8 K memory--only one normalized $^{87}Sr/^{86}Sr$ ratio is computed for each block from the mean $^{87}Sr/^{86}Sr$ and $^{88}Sr/^{86}Sr$ ratios ($^{86}Sr/^{88}Sr$ is set to 0.1194). No spiking calculations are included in the program when the sample contains ^{84}Sr.

METHOD OF COMPUTATION

Baselines for each peak are stored as straight lines (slope and intercept, m_b and c_b respectively). These lines join the mean of the measurements at the beginning and end of each block. They have the form

$$y_b = m_b t + c_b .$$

Peak height information is also stored as straight lines (slope and intercept, m_p and c_p respectively). These lines join consecutive measurements of the same peak and are represented by

$$y_p = m_p t + c_p .$$

Peaks are corrected for baselines simply by subtracting the slope and intercept parameters in the following manner,

$$y'_p = m'_p t + c_p'$$

where $m'_p = m_p - m_b$

and $c'_p = c_p - c_b$.

In a strontium analysis, a similar procedure is used to correct the ^{87}Sr peak for the ^{87}Rb, based on the line parameters computed from the ^{85}Rb measurements,

$$y'_{87} = m'_{87} t + c'_{87}$$

where $m'_{87} = m_{87} - \frac{m_{87}}{2.59}$

and $c'_{87} = c_{87} - \frac{c_{85}}{2.59}$.

In each switching cycle a ratio is computed for each peak relative to the reference peak. Thus:

$$r(t_{ref}) = \frac{y_p(t_{ref})}{y_{ref}(t_{ref})}$$

$$= \frac{m'_p(t_{ref}) + c'_p}{m'_{ref}(t_{ref}) + c'_{ref}}.$$

An additional ratio for each peak relative to the reference peak is computed at the occurrence of each peak.

$$r(t_p) = \frac{y_p(t_p)}{y_{ref}(t_p)}$$

$$= \frac{m_p'(t_p) + c_p'}{m'_{ref}(t_p) + c'_{ref}}.$$

If there are N switching cycles in a block, a total of 2(N-1) ratios is obtained for each peak per block. This method of computation helps to eliminate the effects of curvature in the decay curve of the ion beam signal, without the redundancy of computing all ratios at the occurrence of each peak.

LOADING AND RUNNING THE PROGRAM FROM PAPER TAPE

After loading the CALL BASIC interpreter, press the STOP switch and check that the data switches are set to X7777. Load the assembled driver tape called "PACKAGE.LDA" in the teletype reader and press RESET and START. The driver tape will be read into core. Restart BASIC via address 00002 on the data switches and load the main control program "NEW" from the teletype reader.

The following conversation should then take place between the operator and the computer. Underlined parameters are typed by the operator. Assume we wish to run a strontium sample spiked with ^{84}Sr.

RUN

PKS? 4	4 main peaks in the switching sequence
?1 ? 3 ? 4 ? 5	Assign ^{84}Sr to #1 channel ^{86}Sr to #3 channel ^{87}Sr to #4 channel ^{88}Sr to #5 channel
TIMES	
? 20 ? 20 ? 20 ? 15	Total times per peak - including skip time Assign 20 seconds to peaks #1, #3 ,and #4 Assign 15 seconds to peak #5
SKIP	
? 4 ? 4 ? 4 ? 3	Ignore first 4 seconds for peaks #1, #3, and #4 Ignore first 3 seconds for peak #5 Total useful time per peak (TIME-SKIP).

B/L S? 0	Type 0 for baselines to be measured both sides of peak
	Type 1 for baselines to be measured below peaks
	Type 2 for baselines to be measured above peaks
RB#? 2	Assign ^{85}Rb to channel #2
	If your reply is 0 the computer assumes this is not a strontium analysis being run
C/S? 5	Any number of CYCLES from 2 to 10 may be requested here
BLKS? 5	Any number of BLOCKS from 1 to 10 may be asked for here

After typing RETURN the computer will begin the analysis.

MODIFICATIONS TO DEMONSTRATE THE OPERATION OF PROGRAM

The following list of modifications below will enable the program to be run with the ordinary BASIC interpreter and without the driver program. A set of synthetic data will be generated having the following properties:

(1) All main peak heights will be measured as 11

(2) All baselines will be measured as 1

(3) The #2 peak will be measured as 3.59

When baselines are subtracted all peaks except #2 will have a height of 10 and that of #2 will be 2.59.

If a strontium analysis is called with ^{85}Rb assigned to channel #2 and the other peaks assigned as described earlier, then the ^{87}Sr peak will be corrected by the amount

$$\left(\frac{1}{2.59}\right) \times {}^{85}Rb = 1$$

$$\therefore {}^{87}Sr \text{ corrected} = 9$$

and $^{87}Sr/^{86}Sr = 0.9$, $^{88}Sr/^{87}Sr = 1.0$

and $^{84}Sr/^{86}Sr = 1.0$ if a spiked run is asked for.

The $^{87}Sr/^{86}Sr$ ratio will be normalized by setting $^{86}Sr/^{88}Sr = 0.1194$

The statements to modify the program are:

```
         710 LET  R8 = 0
         825 LET  X  = 11
         910 LET  R8 = 0
8095 and 8029 LET  Y2 = 0
         876 LET  L(I,J1) = 1
         891 IF  A(I)<>R GOTO 893
         892 LET H(I,J) = 3.95
```

If you wish to observe the complete switching sequence for an analysis, the following PRINT statements will print peak numbers and mean measurements as the analysis proceeds:

```
881 PRINT "881"; A(I); L(I,J1); M(I,J1)
896 PRINT "896"; A(I); H(I,J); Q(I,J)
```

Appendix B shows an input-output listing for a "strontium" analysis using the synthetic data described here.

BASIC SOFTWARE DESCRIPTION

PROGRAM STRUCTURE

		Line Numbers
<u>Initializing</u>		20 - 499
	Dimensions, Set Counters, zeros arrays	
<u>Data Acquisition</u>		
"INPUT"		500 - 698
	Acquires sequence control data for the analysis from the operator (see p. 16 for parameter symbols)	
"PEAK SWITCH"		700 - 740
	Subroutine - does just that.	
"READ"		800 - 885
	Routine for reading counter	
"RUBIDIUM"		900 - 950
	Switches and reads rubidium peaks	
"BASELINE"		1000 - 1080
	Switches and reads baselines	
<u>Data Processing</u>		
"RB LINE"		1140 - 1190
	Slopes and intercepts computed for rubidium peak	
"BASELINE LINES"		1200 - 1298
	Slopes and intercepts computed for baselines	

"PEAK LINES"		1300 - 1450
	Slopes and intercepts computed and stored for peaks	
"RATIOS-REFERENCE"		1600 - 1720
	Ratios for each peak with respect to the reference peak computed at each time of occurrence of the reference peak	
"RATIOS-PEAKS"		1800 - 1870
	A ratio for each peak with respect to the reference peak is computed at each of the times of its own occurrence.	
	Headings printed for ratios; arrays loaded and printed	3120 - 3340
"FPRINT"		3500 = 3960
	Means and standard deviations printed. Individual ratios then filtered by rejecting those with deviations more than 2σ from the mean. Rejected ratios and their times are printed. New means and σs printed. The process is repeated until no ratios lie outside 2σ from the recomputed mean.	
"NORMALIZE"		4000 - 4040
	Normalizes the mean $^{87}Sr/^{86}Sr$ for the block, assuming $^{88}Sr/^{86}Sr = 0.1194$.	

"APRINT" 5000 - 5310

Headings and table printed for mean ratios from each block at end of analysis. Analysis may comprise up to 10 blocks.

"MAIN" 8000 - 8500

Routine for controlling overall sequence of program

PARAMETER SYMBOLS USED IN INPUT SUBROUTINE

On the Nova computer BASIC software, all arrays contain a "zeroth" position, and these positions have been utilized in our program to save memory. Thus, most counting parameters such as N, I, and B2 start at "zero" and are one less than their physical counterparts.

N1	Number of main peaks in the spectrum	
N	is (N1-1)	
A[I]	Peak channel number for the Ith peak	
T[I]	Total time to be spent on Ith peak	
S[I]	Total SKIP time on Ith peak	
B1	Baselines to be measured "Below each peak"	B1 = 1
	"Both sides of each peak"	B1 = 0
	"Above each peak"	B1 = 2
R	Channel number for rubidium-85	
	If R = 0, the analysis is NOT strontium.	
C1	Number of switching cycles per block	
B2	Number of blocks (less one)	

COMPUTATIONS DURING DATA ACQUISITION

During data acquisition, all baseline and peak height measurements are accumulated and stored as mean values of the appropriate groups of incoming data--together with their mean times. They are stored as L(I,J1) and M(I,J1) respectively for baselines and H(I,J), Q(I,J) respectively for peak heights and times (lines 875-895 in the program). Note that accumulation does not start until after points to be "skipped" at the beginning of each group have been ignored.

BASELINE AND RUBIDIUM CORRECTIONS TO PEAK HEIGHTS

After the acquisition of data for a block is completed, parameters (slope and intercept) for lines joining mean baseline measurements for each peak at end of the block are calculated (M1 and L1, lines 1280, 1285). Slope and intercept for each line joining consecutive measurements of the same peak are computed as Q1 and P1 - lines 1320, 1331. These last two parameters are corrected for baseline by subtracting M1 and L1 (lines 1380, 1390) and stored as G(I,K) and H(I,K). I is the peak number (less one) and K is the line number(starting at K = 0). There are C1-1 lines for each peak where C1 is the number of switching cycles within a block.

Rubidium corrections to ^{87}Sr peaks are made in the same manner as baseline corrections - at lines 1360, 1370.

COMPUTATION OF RATIOS

Ratios are calculated for all peaks with respect to the reference peak at each occurrence of the reference peak. (Usually the reference peak channel number is 3--that is, R2 = 2). These ratio calculations are made in subroutine RATIOS-REF (1600 - 1620).

Within each switching cycle, another ratio for each peak with respect to the reference peak is calculated--at the occurrence of that peak (RATIOS-PEAKS, lines 1800-1870).

In each of the above instances two cases arise, depending on whether the peak number is greater than or less than that of the reference peak (that is, whether I>R2 or I<R2). Consider calculation of ratios at the reference peaks in subroutine RATIOS-REF. Let I<R2 so that R2=2 and I=1 as in figure 1. Starting at the FIRST occurrence of the reference peak (C1-1) ratios will be calculated at times Q(R2,J) for J=0 to J=(C1-2). The peak line parameters H(I,J), G(I,J) for the I=1 peak and H(2,J), G(2,J) for the reference peak are used in each calculation. For peaks where I>R2 (I=3 as in figure 1) the calculations are similar except that we start at the SECOND occurrence of the reference peak. Thus, (C1-1) ratios are computed at times Q(2,J1) for J1=(C1-1)

Subroutine RATIOS-PEAK calculates individual ratios at the occurrence of the peaks themselves in a similar manner. (C1-1) ratios are computed for each peak, the first ratio being calculated at the first or second occurrence of that peak--depending on whether I>R2 or I<R2 respectively.

From the two subroutines, a total of 2*(C1-1) ratios are calculated for each peak in the main switching sequence, with respect to the reference peak.



FIG. 1 Diagram for ratio computations in a 5-cycle block using 3 peaks in the main switching sequence.

ASSEMBLY LANGUAGE DRIVER PROGRAM

The complete driver comprises three subroutines labelled "START", "SIG", and "BCNT".

START operates the peak SWITCH and baseline offset relays in the mass spectrometer, under control of the main BASIC program. It selects the relays to be closed from the statement: "710 CALL 0, D, A[I]-1" in which the '0' parameter calls SIG itself. The D parameter (value 8, 9 or 10) determines the selection of baseline below, baseline above or peak top. A[I] is the peak relay number. These last two parameters are read by the assembly language statement GET. START then sets up the necessary bit pattern to operate the relays via the relay driver circuits in the interface unit.

The BASIC statement "825 CALL I, X" calls subroutine SIG and when the main counter is DONE, 6 BCD digits are read from the interface (4 from the A register and 2 from the B register). This string is transferred into the BASIC program as a decimal number X by the statement 'PUTD'. The counter is reset by SIG to permit the next reading to accumulate in the counter.

When the final subroutine BCNT is called by "8095 CALL 2, 1" the interblock print-time register BCTR is initiated. It keeps track of time by counting the number of measurements made by the main counter. At the end of printout between blocks, statement "8233 CALL 2,-1, Y2" stops BCTR. The primary count from BCTR is returned to the BASIC program as parameter Y2 by the assembly language statement PUT.

REFERENCES

Stacey, J. S., Wilson, E. E., Peterman, Z. E., and Terrazas, R., 1971, Digital recording of mass spectra in geologic studies--I: Canadian Jour. Earth Sci., v. 8, no. 3, p. 371-377.

Stacey, J. S., Wilson, E. E., and Terrazas, R., 1972, Digital recording of mass spectra in geologic studies--II: Canadian Jour. Earth Sci., v. 9, no. 7, p. 824-834.

APPENDIX A

BASIC PROGRAM LISTING

INITIALIZE .

```
20 DIM A(5),T(5),S(5),P(5,10),Q(5,10),L(5,5),M(5,5),G(5,10)
 21 DIM H(5,10),W(5,10),E(18,2),B(5),C(5)
25  LET B3=-1
122 FOR I=0 TO 4
123    IF B3>= 0 GOTO  140
125 LET A(I)=0
126 LET T(I)=0
127 LET S(I)=0
129 LET B(I)=0
130 LET C(I)=0
140 FOR J=0 TO 9
142 LET G(I,J)=0
143 LET W(I,J)=0
146 LET Q(I,J)=0
147 LET H(I,J)=0
148      IF B3<> 0 GOTO  160
149      LET P(I,J)= 0
150      IF J>=4 GOTO  160
156      LET L(I,J)= 0
157      LET M(I,J)= 0
160    NEXT J
165  NEXT I
170 FOR I=0 TO 17
175 FOR K=0 TO 1
180 LET E(I,K)=0
182    NEXT K
183  NEXT I
185 IF B3<>-1 GOTO 8029
190  LET Z2= 0
191  LET J1= 0
192  LET J= 0
196  LET F=-1
197  LET I2= 0
499  GOTO  8000
```

```
500  REM       I/P - - - - - - - - - INPUT
505  PRINT " PKS";
510  INPUT N
515  LET N1=N
516 LET N=N-1
520  PRINT
530 FOR I=0 TO N
540 INPUT A(I)
550  NEXT I
555  PRINT
560  PRINT " TIMES"
570 FOR I=0 TO N
580 INPUT T(I)
590  NEXT I
600  PRINT
610  PRINT " SKIP"
620 FOR I=0 TO N
630 INPUT S(I)
640  NEXT I
650  PRINT
651  PRINT " B/LS";
652  INPUT B1
653  PRINT
655  PRINT " RB#";
656  INPUT R
657  IF R= 0 GOTO  664
658  LET A(N1)=R
660  LET S(N1)=S(N-1)+1
662  LET T(N1)=T(N-1)+8
664  PRINT
665  PRINT " C/S";
670  INPUT C1
675  PRINT
680  PRINT " BLKS";
690  INPUT B2
692  LET B2=B2-1
695  PRINT
698  RETURN
```

```
700  REM        SW- - - - - - - - - - - - - -PEAK SWITCH
705 FOR I=0 TO N
710 CALL 0,D,A(I)-1
720    GOSUB  800
730  NEXT I
740  RETURN
800  REM        RD- - - - - - - - - - - - - -READ
805  LET Y= 0
806  LET Z1=Z2
810  LET K= 0
812  IF D=10 GOTO  820
814  LET U=2+S(I)
815  LET V=8+T(I)
818  GOTO  825
820  LET U=S(I)
822  LET V=T(I)
825 CALL 1,X
828  LET Z2=Z2+1
830  LET K=K+1
840  IF K<=U GOTO  825
850  LET Y=Y+X
860  IF K<V GOTO  825
870  IF D=10 GOTO  890
875 LET L(I,J1)=Y/(K-U)
880  LET M(I,J1)=Z1+(K+U)/2
882  IF F<>-1 GOTO  935
885  RETURN
890  LET H(I,J)=Y/(K-U)
893 IF H(I,J)>=0 GOTO 895
894 LET H(I,J)=0
895  LET Q(I,J)=Z1+(K+U)/2
898  IF F<>-1 GOTO  946
899  RETURN
```

```
900  REM        RB- - - - - - - - - - - - -RUBIDIUM
901  LET I=N1
902  IF J<> Ø GOTO  952
903  LET F= Ø
904  LET D=8
906  LET S1=J1
908  LET J1=F
910 CALL Ø,D,R-1
920  GOTO  8Ø5
935  LET D=1Ø
940  LET S2=J
942  LET J=F
944  GOTO  91Ø
946  LET J1=S1
948  LET J=S2
949  LET F=-1
950  RETURN
952  LET F=1
954  GOTO  9Ø4
1000  REM        B/LS - - - - - - - - - - -BASELINES
1003  IF J= Ø GOTO  1Ø1Ø
1006  LET J1=J9
1010  IF B1=2 GOTO  1Ø2Ø
1015  LET D=8
1017  GOTO  1Ø3Ø
1020  LET D=9
1030  GOSUB  7ØØ
1035  LET J1=J1+1
1040  IF B1<> Ø GOTO  1Ø7Ø
1050  LET D=D+1
1060  GOSUB  7ØØ
1062  LET J1=J1+1
1070  IF J<> Ø GOTO  1Ø8Ø
1075  LET J9=J1
1080  RETURN
```

```
1140  REM       RB/L- - - - - - - - - - - - -RB LINE
1145  IF R= 0 GOTO  1200
1150  LET I=N1
1155  LET K= 0
1160  GOTO  1268
1170  LET S1=M1
1175  LET S2=L1
1178  IF B3= 0 GOTO  1320
1180  GOTO  1316
1185  LET S1=Q1-S1
1190  LET S2=P1-S2
1200  REM       B/L LS- - - - - - - - - - - -BASELINE LINES
1201  IF R= 0 GOTO  1210
1203  IF B1<> 0 GOTO  1210
1205  LET L(N, 0)=L(N,1)
1207  LET L(N,2)=L(N,3)
1210 FOR I=0 TO N
1230    IF B1<> 0 GOTO  1268
1235    LET K= 0
1240 FOR K1=0 TO 2 STEP 2
1242      IF B3= 0 GOTO  1250
1247      LET K1=2
1248      LET K=1
1250      LET L(I,K)=(L(I,K1)+L(I,K1+1))/2
1255      LET M(I,K)=(M(I,K1)+M(I,K1+1))/2
1260      LET K=1
1265    NEXT K1
1268    IF B3= 0 GOTO  1280
1270    LET L(I, 0)=B(I)
1275    LET M(I, 0)=C(I)
1280    LET M1=(L(I, 0)-L(I,1))/(M(I, 0)-M(I,1))
1285    LET L1=L(I, 0)-M1*M(I, 0)
1290    LET B(I)=L(I,1)
1295    LET C(I)=M(I,1)
1298    IF I=N1 GOTO  1170
1300    REM       PK/LS - - - - - - - - - - -PEAK LINES
1310 FOR K=0 TO C1-2
1312      IF R= 0 GOTO  1320
1314      IF B3= 0 GOTO  1320
1316      LET H(N1, 0)=H8
1318      LET Q(N1, 0)=Q8
1320      LET Q1=(H(I,K)-H(I,K+1))/(Q(I,K)-Q(I,K+1))
1330      LET P1=H(I,K)-Q1*Q(I,K)
1331      IF I<>N1 GOTO  1340
1332      LET H8=H(N1,1)
1333      LET Q8=Q(N1,1)
1335      GOTO  1185
1340      IF R= 0 GOTO  1380
1350      IF I<>N-1 GOTO  1380
1360      LET Q1=Q1-S1/2.59
1370      LET P1=P1-S2/2.59
1380      LET G(I,K)=Q1-M1
1390      LET H(I,K)=P1-L1
1395 NEXT K
1399  NEXT I
```

```
1410  IF R= Ø GOTO  1450
1420  IF N<>3 GOTO  1450
1430  LET R2=1
1440  GOTO  1600
1450  LET R2= Ø
1600  REM      R/RF - - - - - - - - - - - -RATIOS REFERENCE
1620 FOR I=Ø TO N
1630    IF I=R2 GOTO  1870
1650    FOR J= Ø TO C1-2
1660      IF I<R2 GOTO  1700
1670      LET J1=J+1
1690      GOTO  1715
1700      LET J1=J
1715      LET H1=G(R2,J)*Q(R2,J1)+H(R2,J)
1720      LET H2=(G(I,J)*Q(R2,J1)+H(I,J))/H1
1800      REM      R/PKS
1815      IF I>R2 GOTO  1840
1820      LET J2=J+1
1830      GOTO  1848
1840      LET J2=J
1848      LET H1=G(I,J)*Q(I,J2)+H(I,J)
1850      LET W(I,J)=H1/(G(R2,J)*Q(I,J2)+H(R2,J))
1855      LET H(I,J)=H2
1860    NEXT J
1870  NEXT I
3120  FOR I= Ø TO N
3140    IF I=R2 GOTO  3950
3150    PRINT
3155    PRINT
3160    PRINT A(I);"/";A(R2);"RATIOS-";B3+1
3177    PRINT
3178    LET K= Ø
3180    FOR J= Ø TO C1-2
3190      IF I>R2 GOTO  3280
3210      LET E(K, Ø)=H(I,J)
3220      LET E(K,1)=Q(R2,J)
3230      LET E(K+1, Ø)=W(I,J)
3240      LET E(K+1,1)=Q(I,J+1)
3250      PRINT E(K, Ø);E(K,1)
3260      PRINT E(K+1, Ø);E(K+1,1)
3270      GOTO  3330
3280      LET E(K, Ø)=W(I,J)
3290      LET E(K,1)=Q(I,J)
3300      LET E(K+1, Ø)=H(I,J)
3310      LET E(K+1,1)=Q(R2,J+1)
3320      GOTO  3250
3330      LET K=K+2
3335    NEXT J
3340    PRINT
```

```
3500    REM         FPT- - - - - - - - - - -FPRINT
3502    LET K1=2*(C1-1)-1
3505 LET J=-1
3511    LET S5= 0
3512    LET T5= 0
3513    LET S7= 0
3520    FOR K= 0 TO K1
3530      LET S5=S5+E(K, 0)
3540      LET T5=T5+E(K,1)
3550    NEXT K
3555 LET K=K1+1
3560    LET S5=S5/K
3570    LET T5=T5/K
3580    FOR K= 0 TO K1
3590      LET S6=E(K, 0)-S5
3610      LET S7=S7+S6*S6
3640    NEXT K
3650    LET S6= SQR (S7/K1)
3655 IF B3>B2 GOTO 5080
3657 PRINT S5;S6;T5
3660 PRINT
3662 PRINT
3665 IF S6=0 GOTO 3925
3670 IF J=K1 GOTO 3925
3680 IF K1<5 GOTO 3925
3790    FOR K= 0 TO K1
3800      LET D5=E(K, 0)-S5
3810      IF  ABS (D5)<2*S6 GOTO  3860
3830      PRINT E(K, 0);E(K,1);D5;S6
3840 PRINT
3850      GOTO  3900
3860      LET J=J+1
3870      IF K=J GOTO  3900
3880      LET E(J, 0)=E(K, 0)
3890      LET E(J,1)=E(K,1)
3900    NEXT K
3910 IF K1=J GOTO 3925
3915 LET K1=J
3920 GOTO 3511
3925    LET P(I,B3)=S5
3940    LET Q(I)=S6
3945    LET W(I)=T5
3950  NEXT I
3960  RETURN
4000  REM         NORM- - - - - - - - - - - - - - -NORMALIZE
4010 LET P(N1,B3)=P(1,B3)*2/(1+.1194*P(2,B3))
4015  PRINT "87/86-N IS";P(N1,B3)
4020  PRINT
4025  PRINT
4030  PRINT
4040  RETURN
```

```
5000 REM APT - - - - - - - - - - - - - - - - - - - - - - - - -APRINT
5001 PRINT
5002  PRINT
5003 PRINT,
5004  LET N2=N
5005  IF R= 0 GOTO  5010
5006  IF N=3 GOTO  5010
5007  LET N2=N1
5010  FOR I= 0 TO N2
5020    IF I=R2 GOTO  5110
5030    FOR K2= 0 TO B2
5040      LET E(K2, 0)=P(I,K2)
5050    NEXT K2
5060    LET K1=B2
5070    GOTO  3505
5080    LET Q(I)=S6
5090    LET H(I)=S5
5095    IF R= 0 GOTO  5105
5096    IF I2=2 GOTO  5110
5098    IF N=2 GOTO  5102
5100 PRINT"84/86","87/86","88/86",
5101    GOTO  5109
5102 PRINT"87/86","88/86","87/86-N",
5103    GOTO  5109
5105 PRINT A(I);"/";A(R2),
5109    LET I2=2
5110  NEXT I
5115  IF I>=5 GOTO  5119
5117  PRINT
5119  PRINT
5135  FOR K= 0 TO B2
5137 PRINT,
5140    FOR I= 0 TO N2
5153      IF I=R2 GOTO  5170
5162 PRINT P(I,K),
5170    NEXT I
5175    IF I>=5 GOTO  5190
5180    PRINT
5190  NEXT K
5201  PRINT
5202  PRINT " MEANS",
5210  FOR I= 0 TO N2
5225    IF I=R2 GOTO  5240
5230    PRINT H(I),
5240  NEXT I
5245  IF I>=5 GOTO  5255
5250  PRINT
5255  PRINT " ST.DEVS.",
5260  FOR I= 0 TO N2
5275    IF I=R2 GOTO  5290
5280    PRINT Q(I),
5290  NEXT I
5300  PRINT
5310  RETURN
```

```
8000  REM          MAIN- - - - - - - - - - - - -MAIN
8002  IF B3<>-1 GOTO  8010
8004  LET B3= 0
8010  GOSUB  500
8020  GOSUB  1000
8025 FOR B3=0 TO B2
8026 IF B3=0 GOTO 8032
8027 GOTO 122
8029 CALL 2,-1,Y2
8030 LET Z2=Z2+Y2
8032 IF R=0 GOTO 8045
8035    GOSUB  900
8045      LET D=10
8055 FOR J=0 TO C1-1
8060          GOSUB  700
8070        NEXT J
8080        IF R= 0 GOTO  8090
8085        GOSUB  900
8090        GOSUB  1000
8095        CALL 2,1
8100        GOSUB  1140
8105 IF R=0 GOTO 8130
8110 IF N=3 GOTO 8130
8115 GOSUB 4000
8130 NEXT B3
8132 LET B3=B3+1
8135 IF B2=0 GOTO 8145
8140 GOSUB 5000
8145 FOR J=1 TO 10
8150 PRINT
8155 NEXT J
8160 GOTO 25
8165 END
```

APPENDIX B

INPUT - OUTPUT LISTING

FOR A

"STRONTIUM" ANALYSIS

WITH "SYNTHETIC DATA".

```
RUN
PKS?3 - - - - - - - - - - 3 peaks in main switching sequence

?3                           86Sr → #3
?4    -Peak channel Nos.     87Sr → #4
?5                           88Sr → #5

TIMES - - - - - - - - - - for each peak
?15
?18
?10

SKIP  - - - - - - - - - - for each peak
?4
?4
?3

B/LS?0   - - - - - - - - - baselines - both sides of each peak

RB#?2    - - - - - - - - - 85Rb → #2 peak channel

C/S?3 - - - - - - - - - - 3 switching cycles

BLKS?2  - - - - - - - - - 2 blocks
(1) (2) (3)  (4) - - - - - - - - - - - - - - ► Column No.
881  3   1   14.5  ┐                                 ↓
881  4   1   39    │
881  5   1   60.5  │ Baselines                 (1) Print statement #
881  3   1   81.5  │ Main Peaks
881  4   1   106   │                           (2) Peak channel #
881  5   1   127.5 ┘
881  2   1   154.5 ┐ Peak and Baseline         (3) Mean measurements
896  2   3.59  183.5 ┘  Rb                        ┌ 1 assigned to Baselines
896  3   11  203.5 ┐                              │
896  4   11  220   │                              │
896  5   11  233.5 │           Synthetic data ────┤ 11 assigned to main peaks
896  3   11  246.5 │                              │
896  4   11  263   │ Main Peaks                   └ 3.59 assigned to 85Rb
896  5   11  276.5 │
896  3   11  289.5 │                           (4) Mean times for peak
896  4   11  306   │
896  5   11  319.5 ┘                               measurements.
881  2   1   343.5 ┐ Peak and Baseline
896  2   3.59  372.5 ┘  Rb
881  3   1   397.5 ┐
881  4   1   422   │
881  5   1   443.5 │ Baselines
881  3   1   464.5 │ Main Peaks
881  4   1   489   │
881  5   1   510.5 ┘
```

```
 4 / 3 RATIOS- 1
RATIO   TIME
  .9   220
  .9   246.5
  .9   263
  .9   289.5

  .9   .5376993E-9   254.75
Mean    Standard      Mean
Ratio   Deviation     Time


 5 / 3 RATIOS- 1

 1   233.5
 1   246.5
 1   276.5
 1   289.5

 1   0   261.5


87/86-N IS 1.608004
```

2nd BLOCK STARTS

```
881 2  1   537.5      Peak and Baseline
896 2  3.59  566.5      Rb
896 3  11   586.5
896 4  11   603
896 5  11   616.5
896 3  11   629.5     Main Peaks
896 4  11   646
896 5  11   659.5
896 3  11   672.5
896 4  11   689
896 5  11   702.5
881 2  1   726.5      Peak and Baseline
896 2  3.59  755.5      Rb
881 3  1   780.5
881 4  1   805
881 5  1   826.5      Baselines
881 3  1   847.5      Main Peaks
881 4  1   872
881 5  1   893.5
```

4 / 3 RATIOS- 2

```
.9   603
.9   629.5
.9   646
.9   672.5

.9   .5376993E-9   637.75
```

5 / 3 RATIOS- 2

```
1   616.5
1   629.5
1   659.5
1   672.5

1   0   644.5
```

87/86-N IS 1.608004

FINAL TABLE OF RATIOS

	87/86	88/86	87/86-N	
	.9	1	1.608004	---------- 1st Block
	.9	1	1.608004	---------- 2nd Block
MEANS	.9	1	1.608004	
ST. DEVS.	0	0	0	

PKS? ------------- START NEXT ANALYSIS

APPENDIX C

ASSEMBLY LANGUAGE DRIVER

LISTING

```
;PERSEUS COMPUTING AND AUTOMATION (PTY) LTD
;P. O. BOX 3987, PRETORIA, SOUTH AFRICA
;COPYRIGHT (O) 1971.

;SYMBOL DEFINITIONS FOR 'CALL' PACKAGE
;(NO HIGH SPEED READER OR PUNCH)
;USE WITH SINGLE-USER BASIC KEY-B3 AND WHEN ASSEMBLING
   DRIVER PROGRAM.


000300          DUSR  BUS=300
006156          DUSR  WATT=JSR @156
007620          DUSR  MASK=7770
007771          DUSR BEGIN=10127
003170          DUSR  INIT=3170
007000          DUSR   GET=JSR @0,2
007001          DUSR   PUT=JSR @1,2
003002          DUSR RETURN=JMP @2,2
007003          DUSR  PUTD=JSR @3,2
007004          DUSR  GETD=JSR @4,2
007005          DUSR  GETF=JSR @5,2
007006          DUSR  PUTF=JSR @6,2

                EOT
000265                ERRE=265        ;ERROR TERMINATE
000267                ERRS=267        ;ERROR CONTINUE
```

DRIVER PROGRAM LISTING.

```
---          000060              BCTR=60
             000060              CTR=60
             000060              DRIVER=60
             000300              .LOC .BUS
00300  010136                    FREE
             007771              .LOC BEGIN
07771  000003                    3
07772  007775                    START
07773  010033                    SIG
07774  010124                    BCNT
07775  037000   START:           GET
07776  054430                    STA 3,STORE
07777  102520                    SUBZL 0,0
10000  124000                    COM 1,1
10001  125405   LOOP:            INC 1,1,SNR
10002  000403                    JMP DONE
10003  101120                    MOVZL 0,0
10004  000775                    JMP LOOP
10005  024422   DONE:            LDA 1,CURR
10006  034422                    LDA 3,LOW
10007  117415                    AND# 0,3,SNR
10010  000412                    JMP BLINE
10011  034420                    LDA 3,HIGH
10012  167400   BACK:            AND 3,1
10013  107000                    ADD 0,1
10014  044413                    STA 1,CURR
10015  065060                    DOA 1,DRIVER
10016  034410                    LDA 3,STORE
10017  175405                    INC 3,3,SNR
10020  000404                    JMP FIN
10021  000754                    JMP START
10022  034406   BLINE:           LDA 3,LOW
10023  000767                    JMP BACK
10024  024406   FIN:             LDA 1,BGN
10025  003002                    RETURN
10026  000000   STORE:           0
10027  002000   CURR:            2000
10030  000377   LOW:             377
10031  177400   HIGH:            177400
10032  010032   BGN:             .+1
10033  054773   SIG:             STA 3,STORE
10034  034447                    LDA 3,MASK
10035  056456                    STA 3,@DIGIT
10036  060260                    NIOC CTR
10037  063660                    SKPDN CTR
10040  000777                    JMP .-1
10041  060560                    DIAS 0,CTR
10042  000402                    JMP .+2
10043  061560   BREG:            DIBS 0,CTR
10044  024437   NEXT:            LDA 1,MASK
10045  107400                    AND 0,1
10046  010445                    ISZ DIGIT
10047  046444                    STA 1,@DIGIT
10050  101200                    MOVR 0,0
10051  010434                    ISZ CNT1
10052  000776                    JMP .-2
```

```
10053 034431           LDA 3,RST1
10054 054431           STA 3,CNT1
10055 010432           ISZ CNT2
10056 000766           JMP NEXT
10057 034427           LDA 3,RST2
10060 054427           STA 3,CNT2
10061 010427           ISZ CNT3
10062 000761           JMP BREG
10063 034421           LDA 3,RST1
10064 054423           STA 3,CNT2
10065 034421           LDA 3,RST2
10066 054422           STA 3,CNT3
10067 034737           LDA 3,STORE
10070 010076  LEAD:    FEED
10071 020777           LDA 0,LEAD
10072 007003           PUTD
10073 024417           LDA 1,RST
10074 044417           STA 1,DIGIT
10075 003002           RETURN
10076 022415  FEED:    LDA 0,@DIGIT
10077 024412           LDA 1,ANSCI
10100 123000           ADD 1,0
10101 014412           DSZ DIGIT
10102 001400           JMP 0,3
10103 000017  MASK:    17
10104 177774  RST1:    177774
10105 177774  CNT1:    177774
10106 177776  RST2:    177776
10107 177774  CNT2:    177774
10110 177776  CNT3:    177776
10111 000060  ANSCI:   60
10112 010114  RST:     .+2
10113 010113  DIGIT:   STACK
      000010  STACK:   .BLK 10
10124 007000  BCNT:    GET
10125 125415           INC #1,1 SZR
10126 000406           JMP ON
10127 066460           DIC 1,BCTR
10130 102400           SUB 0,0
10131 007000           PUT
10132 060360           NIOP BCTR
10133 000402           JMP OUT
10134 063060  ON:      DOC 0,BCTR
10135 003002  OUT:     RETURN
      101036           FREE=.
      003170           .END   .INIT
```

APPENDIX D

INTERFACE HARDWARE

DIAGRAMS AND CONNECTIONS

USING NOVA GENERAL INTERFACE

BOARD

PEAK SWITCH RELAY CIRCUIT



RL6, RL7 MADE	BELOW PEAK	BIT 9
RL6 MADE	ON PEAK	BIT 11
RL6, RL7 OPEN	ABOVE PEAK	BIT 10 (not used)

RV1 – RV5 1KΩ 10-turn
RV6, RV7 500Ω 10-turn

INCREASE RESISTANCE AND FIELD, CLOCKWISE



GENERAL PURPOSE INTERFACE: DATA REGISTERS (4041)


B-REGISTER
C-REGISTER
DATA 7
DATA 6
DATA 5
DATA 4
DATA 15
DATA 14
DATA 13
DATA 12
DATA 11
DATA 10
DATA 9
DATA 8
DIGIT 5 BIT 1
DIGIT 5 BIT 2
DIGIT 5 BIT 3
DIGIT 5 BIT 4
DIGIT 6 BIT 1
DIGIT 6 BIT 2
DIGIT 6 BIT 3
DIGIT 6 BIT 4
7401
7400
7493
4.7K
+5V
DATIB
DATIC
10/0602
PRT CMD
DATOC
IOPLS



□ WIRE WRAP PIN

○ CONNECTOR PIN

GENERAL PURPOSE INTERFACE: BUSY, DONE, INTERRUPT (4040)



□ WIRE WRAP PIN

○ CONNECTOR PIN

GENERAL PURPOSE INTERFACE: BUS SIGNALS AND DEVICE SELECTION (4040)

TABLE 1--CONNECTIONS FOR INPUT REGISTER-A GENERAL PURPOSE INTERFACE BOARD.

SIGNAL NAME	WRAP PINS		SLOT 9 PIN #	MOTHER BOARD EDGE CONNECTOR PIN #
M.S.B. Digit 4 Ctr.	12A	41	A67	11
	12	49	A73	8
	11	49A	A75	7
L.S.B. Digit 4 Ctr.	10A	57	A79	19
M.S.B. Digit 3 Ctr.	22	67	A81	20
	21	68A	A83	22
	20A	68	A85	24
L.S.B. Digit 3 Ctr.	20	69	A87	26
M.S.B. Digit 2 Ctr.	28A	70A	A89	27
	28	70	A91	3
	27	71	B6	29
L.S.B. Digit 2 Ctr.	26A	72A	B11	30
M.S.B. Digit 1 Ctr.	36	72	B13	31
	36A	76	B15	32
	35	83	B19	33
L.S.B. Digit 1 Ctr.	34	89	B25	35

TABLE 3--CONNECTIONS FOR OUTPUT REGISTER GENERAL PURPOSE INTERFACE BOARD.

SIGNAL NAME	WRAP PIN FROM	TO	SLOT 9 PIN #	MOTHER BOARD EDGE CONNECTOR PIN #
RLY 1	24	131	B34	38
RLY 2	24A	132A	B36	39
RLY 3	25	132	B38	40
RLY 4	26	133	B40	41
RLY 5	23	134A	B48	42
RLY 6 & 7	40	98A	B49	43
RLY 6	76A	135	B53	46